CHAPTER 4: Micro-analysis of the trade effects of non-tariff measures

TABLE OF CONTENTS

A. Overview and learning objectives

Most existing evidence on the impact of non-tariff measures (NTMs) on trade flows is based on sector level observations. However, a precise understanding of their impact would require more disaggregated information that is at the firm level. For instance, working with firm level data allows identify heterogeneous patterns of NTMs effects on production and trade flows. Ideally the all universe of firms should be included in the empirical analysis. In this context it would be possible to identify the effects of any measure on all trade margins. In practice, the most accessible data are trade data as collected by customs. This implies that only firms which have been able to reach some foreign market can be considered. This creates a *de facto* selection bias and limits the scope of the empirical set up. Nonetheless this is still an important first step in a more detailed assessment of NTMs effects. This Chapter discusses the details of such assessment exercises and reviews the existing empirical results. Based on real firm level export data some practical assessment is also included.

In this chapter you will learn how to assess the impact of NTMs on the margins of trade using firm-level observations.

B. Analytical Tools

1. Firm level analysis

To date, little evidence has been provided in the literature on the effects of NTMs on individual firms and their export decisions. NTMs may affect both the extensive margin (i.e. the probability of export) and the intensive margin of trade (i.e. the volume of trade per firm). Overall, firm-level studies show a negative effect of NTMs on both trade margins. Two potential reasons may explain this result. First, firm-level analyses focus on restrictive NTMs. Second, NTMs may be particularly trade restrictive for small firms and outsourcing firms.

NTMs, and in particular SPS measures and TBTs, impose product characteristics that are usually synonymous with additional inputs and consequently, an increase in the cost of producing and shipping the good facing these regulations. The implementation of an NTM may represent an additional fixed cost, as it may require some product or production process adaptation. For instance, a TBT may impose a requirement to use environmentally friendly equipment. If the producer does not possess this type of equipment, buying or even renting it will increase the cost of production independently of the quantity produced. That is, it will increase the fixed cost of production.

An increase in fixed costs may translate into an increase in the cost of entry into the good's export market. Those producers that are unable to buy the required equipment in order to meet the NTM's requirements will not be able to export at all. Firms with lower productivity are more likely to be excluded from the export market.

NTMs may also represent a variable cost whenever they affect the access per se to an export market, as in the case of shipment inspections. In that context, an NTM would affect domestic

and foreign producers differently. And, although it would equally affect exporters of different sizes, it would have a lesser effect on exporters of high-quality products as these are characterized by higher mark-ups. All in all, exporters' and products' characteristics are crucial in determining the effect of shocks from NTM-related fixed and variable costs.

Although there are clear theoretical results on the respective effects of changes in fixed cost and variable costs of producing and exporting, it is difficult to test these findings empirically. The main reason, as usual, is accessibility to relevant and usable data.

Which kind of data can be used to investigate these issues? The quantification of the fixed costs induced by product adaptation can be calculated through surveys of firms. To examine whether the market share of large exporters increases, these survey data can be merged with customs data providing the value and quantity exported by each firm to each destination and for each product. Similarly, the potential exclusion from the international market of less productive firms may be analysed using customs data. If a firm exports a given product to a given destination in a given year and not the year after, then one can conclude it has exited from the export market. Variable costs related to the systematic inspections of shipments can also be measured through firms' surveys. The heterogeneous impact of these inspection costs on domestic and foreign producers is however hardly observable. To study this differentiated effect, one needs information on domestic value added and imports based on the same product classification. On the other hand, customs data are useful to analyse whether NTM-related variable costs equally affect exporters of different sizes. Finally, to observe whether exporters of high-quality products are less affected by NTM-related variable costs, one can check whether the quality and unit value of exported products differ across exporters.

Previous theoretical insights assumed that NTMs are necessarily enforced. However, it is not always the case, and this may affect the impact of NTMs from both a theoretical and empirical points of view. Therefore we may want to include some indication of the uncertainty of NTM enforcement to identify more precisely the effects of NTMs in an empirical set-up. Relevant information can be found in firms' surveys containing questions regarding the procedural obstacles they face. "Border alerts," such as those reported by the European Rapid Alert System for Food and Feed, could also be used to establish some specific identification strategy. The existence of border alerts protocols can be related explicitly to the probability of enforcement.

2. Impact of non-tariff measures on firms' export decisions

Fontagné *et al.* (2015) investigate the impact of these NTMs on individual firms' export decisions (whether or not to export, and how much to export) and pricing behaviour in export markets. Their focus is on SPS measures, and more precisely on "restrictive SPS measures," i.e. those that have been raised as concerns in the WTO SPS Committee. They work with French firm export data at the HS four-digit level and cover 10 years (1996–2005). Their sample is restricted to extra-European export flows and to firms that export a certain product to a certain market for at least four years within the 1996–2005 period. The authors also exclude small destinations (i.e. destinations below the median in terms of export flows received from France). Their estimations control for firm size and firm visibility on the export markets.

Results suggest that SPS concerns reduce the probability of export by 4 per cent and the number of exporting firms by 8 per cent. Results remain unchanged if SPS measures in the form of an import ban are excluded from the sample. SPS concerns also increase the probability that firms exit a given product-destination market by 2 per cent. Adaptation costs may be too high for some firms, which are excluded from the market. It does not necessarily mean that these firms stop exporting in general; they may re-orient their exports to another product destination market. On the other hand, tariffs do not affect the exit probability. Export costs associated with SPS measures seem therefore to be more important than variable export costs (e.g. tariffs). At the intensive margin, SPS concerns reduce the export values for firms staying in the market by 18 per cent. Overall, the quantification exercise by the authors suggests an expected lost sales figure from SPS measures at the firm level of 43 per cent. The extensive margin accounts for one-third of this total effect and the intensive margin for two-thirds. The effects of SPS concerns are however heterogeneous across firms, and big players are less affected than small firms at both the extensive and intensive margins. In other words, because of SPS concerns, small firms are more likely to participate less in the export market, to exit this export market, and to export smaller trade volumes than large firms. Last but not least, SPS concerns increase average prices of exported goods by 6 to 9 per cent. This last result can be explained by the fact that SPS measures increase production costs and act as barriers to entry. Thus, market shares are redistributed about remaining firms, which increase their prices.

Fernandes *et al.* (2015) focus on the impact of pesticide standards on firm exports of agricultural products across countries and time. Their results show that pesticide standards significantly influence the foreign market access of affected products. More restrictive standards (i.e. imposing a lower legally tolerated level of a pesticide residue in products) in the importing country, relative to the exporting country, lower the probability that firms will export as well as their export values and quantities. They also find evidence of heterogeneous effects among exporters. Smaller exporting firms are more negatively affected than larger ones, in their market entry and exit decisions, by the relative stringency of standards.

Disdier *et al.* (2018) show that the introduction of SPS measures and TBTs in foreign markets increase the probability of exporting, as well as the value of exports, of the most productive French exporters, while reducing the exporting probability of the least productive French firms.

Fugazza *et al.* (2018) combine data on tariffs and NTMs imposed by Latin American importers with firm-level export data from Peruvian customs during the period 2000-2014 to examine the impact of market-access barriers in Latin America on Peruvian exporters. While Disdier *et al.* (2018) rely on a cross-section of SPS measures and TBTs imposed by France's trading partners, this dataset allows to identify the impact of the introduction and withdrawal of different types of NTMs. They find that the average impact varies depending on the type of trade barrier. Some market-access barriers, such as tariffs and TBTs, hurt the average exporter more than other measures, such as price controls. They also find that the impact of market-access barriers is not homogeneous across exporters. Small exporters are affected by market-access barriers more negatively than large exporters are. Interestingly, very large exporters tend to benefit, rather than lose, from the imposition of more restrictive market-access barriers in destination markets. The case of Peruvian exports to other Latin American Integration Association (LAIA) partners is interesting because of two apparently contradicting trends. While the share of Peruvian exports directed to LAIA countries

has been growing since 2000, the number of Peruvian exporters to the region has been declining during the same period. While the intensification of exports to LAIA countries could be associated with the economic and trade integration process at work in the region over the last fifteen years, the increasing concentration of firms in the export sector is puzzling. The econometric analysis reveals that while tariffs were being bilaterally reduced within LAIA, there was a growing implementation of NTMs, and in particular technical regulations, which hurt small Peruvian exporters, while benefitting large exporters to the region.

3. Impact of non-tariff measures on export diversification

Firm-level data can also be used to examine export diversification. Using data for 619 firms from 17 developing countries exporting to five developed ones, Chen *et al.* (2008) study firms' export performances over two dimensions: (a) their export propensity (i.e. their overall export share), and (b) their market diversification (i.e. the number of export markets served). The authors use data on NTMs based on surveys in which firms are asked whether technical regulations affect their ability to export products.

Results highlight that TBTs affect developing exporters' entry into developed markets; firms claiming that technical regulations are obstacles to trade also export to fewer markets. More interestingly, Chen *et al.* (2008) show that each type of regulation has a differentiated trade impact. Quality standards and labelling requirements increase firms' average export propensity and market diversification, while certification procedures reduce firms' average export diversification.

Shepherd (2015) studies the impact of NTMs and their harmonization on export diversification. His focus is on textiles, clothing, and footwear – three sectors associated with the early stages of industrialization in many countries. Using information on European regulations from the World Bank's European Union Standards Database and detailed European import data between 1995 and 2003, Shepherd shows that NTMs negatively affect the export diversification of developing countries. For an average developing country, the elasticity of export variety with respect to the total number of European standards is –0.6. On the other hand, harmonization mitigates the effect. A one percentage point increase in the proportion of European NTMs harmonized with international NTMs is associated with a growth in export variety of 0.8 per cent. Thus, international NTM harmonization can be an effective way of supporting developing countries' export diversification.

C. Application

Estimating trade effects of non-tariff measures at the firm level

This application uses firm-level data for some unspecified country. It examines the impact on firms' exports of SPS measures and TBTs implemented by the European Union (14 countries; Luxembourg is missing). SPS and TBT data are from the dataset described in chapter 2, B.1 and the firm data are from a developing country. Firm data are available for various years, but NTM data are cross-section data. We therefore focus on 2011 and run cross-section estimations.

(a) Download the data

Both the firm-level data and NTM data are available on the UNCTAD website.[39]

(b) Open the data into Stata and finalize the firm-level dataset

The data should first be downloaded and then be opened in Stata. The programme below finalizes the construction of the dataset by merging firm and NTM data. We first restrict our sample of firms' exports to 2011 and clean the data by dropping incorrect product codes and domestic trade. We also keep only export flows to the European Union. Finally, we expand the dataset of firm exports and generate zero export flows. To do so, we create one European destination for each firm product observation. In other words, we assume that for each product exported by a given firm, the flow can be potentially observed for each European destination. If it is not observed, then it is a zero flow with a value and a quantity set to zero. For non-zero observations, we merge our observations with the initial firm data in order to get the value and the quantity exported by the firm.

```
set memory 800M
set matsize 4000

use "FIRMS", clear
rename y year
* For many countries, NTMs data are for 2011. We decide to work with the 2011 exports of firms
keep if year == 2011
* Drop incorrect HS product codes
gen l = length(hs)
tab l
drop if l != 6
drop l
rename hs hs6
rename d ccode_d
* drop domestic trade
drop if ccode_d == "XXX"
rename c ccode_o
rename f firm
order ccode_o firm year hs6 ccode_d
sort ccode_o firm year hs6 ccode_d
count
* Focus on export flows to EU15
gen eu15_d = 1 if ccode_d == "AUT" | ccode_d == "BEL" | ccode_d == "DEU" | ccode_d =="DNK" |
ccode_d == "ESP" | ccode_d == "FIN" | ccode_d =="FRA" | ccode_d == "GBR" | ccode_d == "GRC"
| ccode_d == "IRL" | ccode_d == "ITA" | ccode_d == "NLD" | ccode_d == "PRT" | ccode_d == "SWE"
replace eu15_d = 0 if eu15_d == .
keep if eu15_d == 1
sort firm hs6 ccode_d
save temp_firmdata, replace
```

[39] We use the 2014 version of the NTM dataset. The full dataset can be downloaded from the CEPII website at http://www.cepii.fr/cepii/fr/bdd_modele/presentation.asp?id=28.

```
* Expand the dataset: create for each firm-product combination, all destinations (i.e. the potential number of destinations: 14 EU countries - Luxembourg is missing)
* We first keep one observation per firm-hs6 code
keep firm hs6
bys firm hs6 : gen obs = _n
keep if obs == 1
drop obs
codebook hs6, compact
codebook firm, compact
egen fp = group(firm hs6)
* for each fp, we create 14 observations: one for each potential EU destination
expand 14
bys fp : gen obs_expand = _n


gen ccode_d = "AUT" if obs_expand == 1
replace ccode_d = "DEU" if obs_expand == 2
replace ccode_d = "DNK" if obs_expand == 3
replace ccode_d = "FIN" if obs_expand == 4
replace ccode_d = "FRA" if obs_expand == 5
replace ccode_d = "GBR" if obs_expand == 6
replace ccode_d = "GRC" if obs_expand == 7
replace ccode_d = "IRL" if obs_expand == 8
replace ccode_d = "ITA" if obs_expand == 9
replace ccode_d = "NLD" if obs_expand == 10
replace ccode_d = "PRT" if obs_expand == 11
replace ccode_d = "SWE" if obs_expand == 12
replace ccode_d = "ESP" if obs_expand == 13
replace ccode_d = "BEL" if obs_expand == 14
drop obs_expand fp
sort firm hs6 ccode_d
save temp_firmdata_expand, replace

* Merge with temp_firmdata to get information on exported quantity and value for strictly positive flows
merge firm hs6 ccode_d using temp_firmdata
tab _merge
drop _merge
replace ccode_o = "SEN" if ccode_o == ""
replace year = 2011 if year == .
replace v = 0 if v == .
replace q = 0 if q == .
save database, replace

erase temp_firmdata.dta
erase temp_firmdata_expand.dta
```

(c) Add information on NTMs

We now deal with NTM data. We first restrict our data to (A) SPS measures and (B) TBTs implemented by European countries. We then merge the firm data with the NTM data and define two NTM variables: a simple dummy "Pres_SPSTBT" set to one if the European Union applies at least one SPS measure or TBT on a given HS six-digit product (0 otherwise). The second NTM variable ("num_SPSTBT") represents the total number of SPS measures and TBTs applied by the European Union on a given product.

```
use "Base_Mast_Oct2014", clear
* Focus on SPS and TBTs
keep isor product numA numB PresA PresB
rename isor ccode_d
gen eu15_d = 1 if ccode_d == "AUT" | ccode_d == "BEL" | ccode_d == "DEU" | ccode_d == "DNK" |
ccode_d == "ESP" | ccode_d == "FIN" | ccode_d == "FRA" | ccode_d == "GBR" | ccode_d == "GRC"
| ccode_d == "IRL" | ccode_d == "ITA" | ccode_d == "NLD" | ccode_d == "PRT" | ccode_d =="SWE"
replace eu15_d = 0 if eu15_d == .
keep if eu15_d == 1
codebook ccode_d, compact
count
gen hs6 = string(product)
gen l = length(hs6)
tab l
replace hs6 = "0"+ hs6 if l == 5
drop product l
sort ccode_d hs6
save temp_NTM, replace

* Merge firm-level data with NTM data
use database, clear
sort ccode_d hs6
merge ccode_d hs6 using temp_NTM
tab _merge
drop if _merge == 2
drop _merge
replace numA = 0 if numA == .
replace numB = 0 if numB == .
replace PresA = 0 if PresA == .
replace PresB = 0 if PresB == .
* Define two variables for SPS and TBTs
* First variable: a simple dummy
gen Pres_SPSTBT = 1 if PresA == 1 | PresB == 1
replace Pres_SPSTBT = 0 if Pres_SPSTBT == .
* Second variable: the total number of SPS and TBTs at the product-importer level
gen num_SPSTBT = numA + numB
drop PresA PresB numA numB
save database, replace

erase temp_NTM
```

(d) Run the estimations using a simple dummy variable for NTMs

Below we run our first set of estimations. To keep zero flows in our regressions, we use the Poisson estimator. Our estimations also include various sets of fixed effects: HS two-digit, firm, and importing-country fixed effects to control for all unobservable characteristics at the sector, firm, and destination levels that may influence the bilateral exports of firms. Note that our sample contains just one exporting country and therefore the importer fixed effect is equivalent to a country pair fixed effect.

In this first set of estimations, we just include a simple dummy for NTMs controlling for the presence versus the absence of SPS measures or TBTs at the product destination level. We consider three different dependent variables: the value of the export flow (regressions (1) and (2)); the exported volume (regressions (3) and (4)); and the unit value of the flow (regression (5)). Regressions (1) and (3) include all flows, while regressions (2), (4), and (5) are restricted to strictly positive export flows.

```
use database, clear
gen hs2 = substr(hs6, 1 , 2)

* Dependent variable: exports value
* Regression #1: all flows
xi: poisson v Pres_SPSTBT i.hs2 i.ccode_d i.firm, difficult robust
* Regression #2: only strictly positive exported values
xi: poisson v Pres_SPSTBT i.hs2 i.ccode_d i.firm if v > 0, difficult robust

* Dependent variable: exports quantity
* Regression #3: all flows
xi: poisson q Pres_SPSTBT i.hs2 i.ccode_d i.firm, difficult robust
* Regression #4: only strictly positive exported volumes
xi: poisson q Pres_SPSTBT i.hs2 i.ccode_d i.firm if q > 0, difficult robust

* Dependent variable: unit value of exports
* Regression #5: only strictly positive unit values
gen uv = v / q
xi: poisson uv Pres_SPSTBT i.hs2 i.ccode_d i.firm if uv > 0, difficult robust
```

(e) Run the estimations controlling for the number of NTMs

We replicate the first set of estimations, but instead of using a simple dummy for NTMs, we now include the total number (in logs) of SPS measures and TBTs notified by European importers on each HS six-digit product. We therefore deal only with importer product observations for which this number is strictly positive.

```
* Log number of SPS/TBT measures at the importer-product level
gen lnum_SPSTBT = ln(num_SPSTBT)

* Dependent variable: value of exports
* Regression #1: all flows
xi: poisson v lnum_SPSTBT i.hs2 i.ccode_d i.firm if num_SPSTBT > 0, difficult robust
* Regression #2: only strictly positive exported values
xi: poisson v lnum_SPSTBT i.hs2 i.ccode_d i.firm if num_SPSTBT > 0 & v > 0, difficult robust

* Dependent variable: quantity of exports
* Regression #3: all flows
xi: poisson q lnum_SPSTBT i.hs2 i.ccode_d i.firm if num_SPSTBT > 0, difficult robust
* Regression #4: only strictly positive exported volumes
xi: poisson q lnum_SPSTBT i.hs2 i.ccode_d i.firm if num_SPSTBT > 0 & q > 0, difficult robust

* Dependent variable: unit value of exports
* Regression #5: only strictly positive unit values
xi: poisson uv lnum_SPSTBT i.hs2 i.ccode_d i.firm if num_SPSTBT > 0 & uv > 0, difficult robust
```

D. Exercises

1. Trade at the firm level
    (i) Preliminaries
        a. Open the datafile "FIRMS.dta"
        b. Drop incorrect HS product codes and domestic trade as in the application
    (ii) Average and total exports
        a. Compute the average size of export relationships by destination and year
        b. Draw the kernel density of average sizes for each year
        c. Compute total exports by destinations and year
        d. Draw the kernel density of total exports for each year

    Hint: use the collapse and kdensity commands

    (iii) Number of exporting firms
        a. Compute the number of exporting firms per year
        b. Identify destinations with the largest number of exporters each year
2. Trade effects of non-tariff measures on firms' exports
    (i) Preliminaries
        a. Open the datafile "database.dta"
        b. Generate an HS 2-digit and HS 4-digit variable
        c. Generate the log number of SPS/TBT measures at the importer-product level

(ii) Trade effects of NTMs

a. Run the estimations controlling for the number of NTMs as in the application

b. Control for HS4 digit fixed effects. Do you see any difference with results obtained in a)?

c. Rerun previous regressions using the *ppml_panel_sg* command

Hints: use the ssc install command. The "ppml_panel_sg" command has been in Yotov et al. (2016)